
第 1章
创建高效 Excel 函数公式

必备技能

Excel 提供了数百个函数，能够在不同的表格中构建各种高
效计算公式，实现复杂、高效的数据处理和数据分析计算。因此，
学习和运用Excel 的重要内容之一，就是掌握Excel 常用函数的
灵活用法与实际应用。
本章主要介绍创建高效公式的必备知识和技能，这些知识和

技能是Excel 函数公式的重要基础，可以说，不了解这些基础知
识和技能，公式出错了都不知道为什么，而且还会频频出错。

1�1 公式的本质，是如何进行计算的逻辑思维

创建公式的目的，是为了计算出需要的结果，这就要求用户必须根据具体表格、
具体问题，选择高效的函数，创建简单、高效的计算公式。
设计公式的过程，其实就是解决问题的逻辑思维，因此，在设计公式之前，首

先要认真阅读表格，梳理逻辑思路，寻找解决方案，并从多个可选的解决方案中选
择一个最高效的解决方案。

1�1�1 仔细阅读表格，寻找解决思路

设计公式不是空谈，更不是套用，应该根据表格的具体情况，仔细阅读，仔细分析，
看看条件是什么，条件在哪里，结果是什么，结果在哪里，然后找出尽可能多的解
决思路。
下面结合两个实际案例介绍如何阅读表格，如何梳理解决问题的逻辑思路，进

而设计高效的计算公式。

案例 1-1

图 1-1 中的示例，要计算每个部门合同工和劳务工的实发工资合计数。

图 1-1　示例数据，计算每个部门合同工和劳务工的实发工资合计数

仔细观察表格，可以这样去思考这个问题：
 ● 任务是什么：计算实发工资的合计数。

 ● 结果在哪里：实发工资在 H 列，这个位置是固定的。

 ● 条件在哪里：B 列判断部门，C 列判断合同类型，这两个条件列位置也是固定的。

 ● 结论：这是两个条件求和问题。

 ● 解决方法：使用多条件求和函数 SUMIFS。

这样，就可以设计如下最简单的求和公式了：
=SUMIFS($H:$H,$B:$B,$K2,$C:$C,L$1)

案例 1-2

图 1-2 中的示例，要求查找指定地区、指定产品的数据。

图 1-2　示例数据：查找指定地区、指定产品的数据

仔细观察表格，也可以这样去思考如下问题：
 ● 任务是什么：查找数据。

 ● 结果在哪里：不清楚在哪里，因为指定的地区和指定的产品是两个变量。

 ● 条件在哪里：这要看解决问题的切入点是什么。

 ● 结论：这是两个条件的查找问题。

 ● 解决方法：工作表实际上是个二维坐标，行和列是坐标。根据解决问题的切入

点（也就是条件判断的切入点），解决方法有很多。

				a）	如果把地区当成条件，把产品当成结果，这个思路就是条件（地区）在
左侧一列，结果（产品）在右侧某列，是从左往右查找数据。因此，首
选 VLOOKUP 函数，但由于指定产品不定，取数的列号也不定，需要使用
MATCH函数来确定这个取数的列号，此时公式为：

    =VLOOKUP(K3,B4:H11,MATCH(K4,B3:H3,0),0)

    b）	如果把产品当成条件，把地区当成结果，这个思路就是条件（产品）在
上面一行，结果（地区）在下面某行，是从上往下查找数据。因此，首
选HLOOKUP 函数，但由于指定地区不定，取数的行号也不定，需要使用
MATCH函数来确定这个取数的行号，此时公式为：

    =HLOOKUP(K4,C3:H11,MATCH(K3,B3:B11,0),0)

				c）	如果把地区和产品都当成两个坐标条件，一个确定行坐标（行号），一个确
定列坐标（列号），那么就需要先使用MATCH 函数确定这两个坐标，再用
INDEX 函数把这个坐标（该单元格）的数据取出来，此时公式为：

    =INDEX(C4:H11,MATCH(K3,B4:B11,0),MATCH(K4,C3:H3,0))

				d）	如果要以数据区域的第一个单元格作为参照点，通过偏移的方法来提取指
定行号、指定列号的数据，那么就需要使用OFFSET 函数了，但仍然需要
使用MATCH函数来确定偏移的行数和列数，此时公式为：

    =OFFSET(B3,MATCH(K3,B4:B11,0),MATCH(K4,C3:H3,0))

    e）	如果想要以整个工作表为坐标系，提取指定行、指定列的数据，则需要使
用 INDIRECT 函数做间接引用，用MATCH 函数来确定数据的行号和列号，
此时公式为：

    =INDIRECT("R"&MATCH(K3,B:B,0)&"C"&MATCH(K4,3:3,0),FALSE)

				f）	本案例中，由于提取的数据类型是数字，还可以将地区和产品作为判断条件，
做两个条件的求和计算，当两个条件都满足时，就是本身数据了，此时公
式为（下面是数组公式）：

    =SUM((B4:B11=K3)*(C3:H3=K4)*C4:H11)

				g）	如果不喜欢使用 SUM函数创建数组公式，可以使用 SUMPRODUCT 函数
来创建普通公式，此时公式为：

    =SUMPRODUCT((B4:B11=K3)*(C3:H3=K4)*C4:H11)

由此可见，这样一个简单表格的数据查找问题，由于解决问题的出发点不一样，
解决的逻辑思路不一样，使用的函数也不一样，设计出来的公式也不一样。而最高
效的公式，就是 VLOOKUP 函数公式和HLOOKUP 函数公式，但从查找灵活性来说，
INDEX 函数公式是最高效的。

1�1�2 梳理逻辑思路，绘制逻辑流程图

任何一个问题本质上都是逻辑判断，满足条件了做什么，满足不了条件做什么，
而函数本身也是各种条件判断，只不过用户看不见函数背后的源代码而已。
设计公式的过程，是梳理逻辑思路的过程，学会绘制逻辑流程图，就是一项非

常重要的技能了，通过逻辑流程图，可以很清楚地知道，每步要做什么，选用什么
函数，参数怎么设置。
前面介绍了如何通过阅读表格来寻找解决问题的逻辑思路，下面介绍如何把这

个思路可视化，如何绘制逻辑思路图，清晰展示解决问题的每一步逻辑与操作。

案例 1-3

图 1-3 中的示例，要根据工龄计算每个人的年休假天数，这个问题需要使用嵌
套 IF 函数解决，因为是多个条件的递层判断。
年休假计算规则如下：

 ● 不满 1 年，0 天。

 ● 满 1 年不满 10 年，5 天。

 ● 满 10 年不满 20 年，10 天。

 ● 满 20 年以上，15 天。

图 1-3　根据工龄计算年休假

对于这个判断处理，可以从小往大依次判断，也就是先判断是否不满 1年，再
判断是否不满 10 年，如此进行下去，直至所有条件判断完毕，此时，可以绘制如图 1-4
所示的逻辑流程图。

图 1-4　逻辑流程图：从小往大依次判断

也可以从大往小依次判断，也就是先判断是满 20 年，再判断是否满 10 年，
如此进行下去，直至所有条件判断完毕，此时，可以绘制如图 1-5 所示的逻辑流	
程图。

图 1-5　逻辑流程图：从大往小依次判断

从 Excel 函数嵌套综合应用来说，当问题不是特别复杂，又是不同函数嵌套的情
况下，也可以绘制函数参数对话框的流程图。
例如，对于案例 1-2 的示例数据，使用 VLOOKUP 函数和MATCH 函数嵌套的参

数对话框逻辑关系如图 1-6 所示，通过这个参数对话框的嵌套逻辑关系，对于问题
解决思路以及函数的嵌套，就非常清晰了。

图 1-6　函数参数对话框的嵌套逻辑关系

�ᐺ本节知识回顾与测验

1. 设计公式的核心是逻辑思维，要认真阅读表格，梳理逻辑思路，这个观点对吗？

2. 如何根据梳理出的逻辑思路来绘制逻辑流程图？

3. 简单函数嵌套公式，如何绘制形象的函数参数对话框式逻辑流程图？请结合

例子自行练习。

4. 对于比较复杂的函数嵌套公式，如何绘制抽象的计算流程式的逻辑流程图？

请结合例子自行练习。

5. 一个简单练习。根据采购量确定折扣率：采购量 1000 以下，不折扣；采购量

1000（含）~2000，折扣为 1.5% ；采购量 2000（含）~5000，折扣为 3% ；采购量 
5000（含）~10000，折扣为 8% ；采购量在 10000（含）以上，折扣为 12%。请绘制

解决这个折扣率的逻辑流程图。

1�2 必备重要技能，高效输入嵌套函数公式

很多人在输入嵌套函数公式时，是手工在单元格中逐个输入函数和参数，结果
按 Enter 键。当提示公式错误时，反回来又重新输入，在这个过程中可能会弹出信
息框提示缺括号少逗号的，这样做好一个公式，往往花费很长时间。
要快速而准确地输入嵌套函数，可以使用函数参数对话框 +名称框方法，也可

以使用分解综合法。下面结合实例，介绍这两种方法的具体应用技能和技巧。

从 Excel 函数嵌套综合应用来说，当问题不是特别复杂，又是不同函数嵌套的情
况下，也可以绘制函数参数对话框的流程图。
例如，对于案例 1-2 的示例数据，使用 VLOOKUP 函数和MATCH 函数嵌套的参

数对话框逻辑关系如图 1-6 所示，通过这个参数对话框的嵌套逻辑关系，对于问题
解决思路以及函数的嵌套，就非常清晰了。

图 1-6　函数参数对话框的嵌套逻辑关系

�ᐺ本节知识回顾与测验

1. 设计公式的核心是逻辑思维，要认真阅读表格，梳理逻辑思路，这个观点对吗？

2. 如何根据梳理出的逻辑思路来绘制逻辑流程图？

3. 简单函数嵌套公式，如何绘制形象的函数参数对话框式逻辑流程图？请结合

例子自行练习。

4. 对于比较复杂的函数嵌套公式，如何绘制抽象的计算流程式的逻辑流程图？

请结合例子自行练习。

5. 一个简单练习。根据采购量确定折扣率：采购量 1000 以下，不折扣；采购量

1000（含）~2000，折扣为 1.5% ；采购量 2000（含）~5000，折扣为 3% ；采购量 
5000（含）~10000，折扣为 8% ；采购量在 10000（含）以上，折扣为 12%。请绘制

解决这个折扣率的逻辑流程图。

1�2 必备重要技能，高效输入嵌套函数公式

很多人在输入嵌套函数公式时，是手工在单元格中逐个输入函数和参数，结果
按 Enter 键。当提示公式错误时，反回来又重新输入，在这个过程中可能会弹出信
息框提示缺括号少逗号的，这样做好一个公式，往往花费很长时间。
要快速而准确地输入嵌套函数，可以使用函数参数对话框 +名称框方法，也可

以使用分解综合法。下面结合实例，介绍这两种方法的具体应用技能和技巧。

1�2�1 同一个函数嵌套：函数参数对话框 +名称框方法

以案例 1-3 的年休假计算为例，输入嵌套 IF 函数的步骤如下。
步骤1单击工具栏中的插入函数按钮，先打开第 1个 IF 函数参数对话框，
然后输入第 1个参数和第 2个参数，如图 1-7 所示。

图 1-7　输入第 1 个 IF 函数的条件和条件成立的结果

步骤2将光标移到第3个参数输入框，然后单击名称框里出现的IF函数，如图1-8所示，
就打开了第 2个 IF 函数参数对话框，然后输入第 2个 IF 函数的判断条件和条件成立
的结果，如图 1-9 所示。

图 1-8　名称框出现的 IF 函数

图 1-9　输入第 2 个 IF 函数的条件和条件成立的结果

步骤3将光标移到第 3个参数输入框，然后单击名称框里出现的 IF 函数，打开第 3个	

IF函数参数对话框，然后输入第3个IF函数的判断条件以及条件成立和不成立的结果，
如图 1-10 所示。
本例是 4种结果，需要使用 3个 IF 函数嵌套，到第 3个 IF 函数要输入全部的 3

个参数。

图 1-10　输入第 3 个 IF 函数的 3 个参数

步骤4单击“确定”按钮，然后往下复制公式，就得到每个人的年休假天数，如图 1-11
所示。最终完成的判断处理公式如下：

=IF(C2<1,0,IF(C2<10,5,IF(C2<20,10,15)))

图 1-11　完成嵌套 IF 函数公式

如果要从大往小进行判断，则判断处理公式为：
=IF(C2>=20,15,IF(C2>=10,10,IF(C2>=1,5,0)))

1�2�2 不同函数嵌套：函数参数对话框 +名称框方法

以案例 1-2 的查找数据为例，输入嵌套 VLOOKUP 函数和MATCH函数
的步骤如下。
步骤1单击工具栏中的插入函数按钮，先打开 VLOOKUP 函数参数对话框，

输入其已确定的第 1个参数、第 2个参数和第 4个参数，其中第 3个参数暂时留空，
如图 1-12 所示。

图 1-12　输入 VLOOKUP 函数已确定的参数

步骤2将光标移到第 3个参数输入框，然后单击名称框下拉列表，选择MATCH函数（如
果未出现MATCH函数，可单击下拉列表底部的“其他函数”，慢慢找到MATCH函数），
如图1-13所示。打开MATCH函数参数对话框，然后输入该函数的3个参数，如图1-14
所示。

图 1-13　从名称框下拉列表中选择 MATCH 函数

图 1-14　输入 MATCH 函数的 3 个参数

步骤3单击“确定”按钮，完成嵌套函数输入，得到查找结果，如图 1-15 所示。

图 1-15　完成不同函数嵌套公式

1�2�3 复杂的嵌套公式：分解综合法

当遇到逻辑更为复杂的不同函数嵌套公式时，最好使用分解综合法，其基本思
路如下：

 ● 先按照最基本的计算单元计算每个项目，并保存在辅助单元格。

 ● 将辅助单元格公式综合到一个公式。

 ● 删除辅助单元格。

案例 1-4

图 1-16 中的示例，要求查找指定产品、指定成本项目、指定年份下各个季度的
成本数据。

图 1-16　复杂的数据查找

仔细阅读表格结构及需求，这是 4个条件的查找：
 ● 条件 1 ：在 A 列匹配产品。

 ● 条件 2 ：在 B 列匹配成本项目。

 ● 条件 3 ：在第 1 行匹配年份。

 ● 条件 4 ：在第 2 行匹配季度。

条件 1和条件 2组合，可以得到指定产品、指定成本项目的行号；条件 3和条件
4组合，则可以得到指定年份、指定季度的列号。
如果直接使用函数参数对话框输入这样的嵌套函数公式，会比较烦琐，也容易

出错，可以使用分解综合法来构建这个表格的数据查找公式。
以指定产品、指定成本项目、指定年份下，1季度的数据查找（单元格Q7）公

式为例，使用分解综合法的具体步骤如下。
（1）分解步骤。

步骤1在辅助单元格Q13 中输入下列公式，确定指定产品的行号：
=MATCH($Q$2,A:A,0)

步骤2在辅助单元格Q14 中输入下列公式，确定指定成本项目的位置（每个产品的
成本项目是一样的，因此任选一个产品下的成本项目区域即可）：

=MATCH($Q$3,$B$3:$B$6,0)

步骤3在辅助单元格Q15 中输入下列公式，确定指定年份的列号：
=MATCH($Q$4,$1:$1,0)

步骤4在辅助单元格Q16中输入下列公式，确定指定季度的位置（每年都是4个季度的，
因此任选一个年份下的季度区域即可）：

=MATCH(P7,$C$2:$F$2,0)

步骤5将确定指定产品的行号和指定成本项目的位置进行计算，就得到指定产品、指
定成本项目的实际行号，保存在辅助单元格Q18 中，公式如下：

=Q13+Q14-1

步骤6将确定年份的列号和指定季度的位置进行计算，就得到指定年份、指定季度的
实际列号，保存在辅助单元格Q19 中，公式如下：

=Q15+Q16-1

步骤7在单元格Q7中输入下列公式，就得到指定条件的数据：
=INDEX($A$1:$N$22,Q18,Q19)

图 1-17 为分解计算过程。

图 1-17　分解计算过程

（2）综合步骤。
上面是公式逐步分解计算出每步的结果，下面将这些步骤综合起来，生成一个

综合公式。
步骤1单元格 Q7 中的公式引用了单元格 Q18 和 Q19 中的数据，将单元格 Q18 和
Q19 的公式字符串套入单元格Q7的公式中，得到如下公式：

=INDEX($A$1:$N$22,Q13+Q14-1,Q15+Q16-1)

步骤2在这个公式中，引用了单元格 Q13、Q14、Q15 和 Q16 中的数据，将单元格
Q13、Q14、Q15 和 Q16 的公式字符串套入单元格Q7的公式中，得到如下公式：

=INDEX($A$1:$N$22,

       MATCH($Q$2,A:A,0)+MATCH($Q$3,$B$3:$B$6,0)-1,

       MATCH($Q$4,$1:$1,0)+MATCH(P7,$C$2:$F$2,0)-1

       )

步骤3将公式往下复制，得到各个季度的数据。
步骤4删除辅助区域，这样就得到了指定产品、指定成本项目、指定年份下各个季度
的成本数据，如图 1-18 所示。

图 1-18　指定产品、指定成本项目、指定年份下各个季度的成本数据

通过以上操作可以看出，这个数据查找问题非常复杂（有 4个条件），其实，如
果对各个条件逐个分解，最后再综合，问题解决的思路既清楚，操作过程又简单。

�ᐺ本节知识回顾与测验

1. 快速、准确输入嵌套函数公式的基本方法是联合使用“函数参数”对话框和

名称框，也就是从名称框中选择要嵌套的函数，打开“函数参数”对话框，输入函

数参数。

2. 输入嵌套函数公式之前，最好绘制逻辑流程图。

3. 针对复杂嵌套函数公式，可以采用分解综合法，也就是先使用每个函数逐步

解决问题，最后再综合为一个公式。

1�3 使用名称创建高效、灵活的数据分析公式

在数据处理和数据分析中，经常会创建一些较长的复杂公式，一看就眼晕，但
是如果使用名称，就可以简化公式，让其变得既逻辑清楚，又容易阅读。

1�3�1 使用名称简化公式

几乎所有的 Excel 对象，如常量、单元格、公式、图形等，都可以定义名称，也
就是为它们起名字。关于名称的定义与使用方法，下面结合一个例子进行说明。

案例 1-5

图 1-19 中的示例，要求从左侧的表格中查找指定业务员、地区、指定产品在各
年的数据，查找公式如下：

=IFERROR(VLOOKUP($M$4,

        IF($M$2=" 业务员 1",$B$3:$J$6,IF($M$2=" 业务员 2",$B$7: 

$J$11,$B$12:$J$14)),

        MATCH(L7,$B$2:$J$2,0)+IF($M$3=" 国内 ",0,4),

        0),

" 没有数据 ")

这个公式很长，主要是 VLOOKUP 函数的第 2个参数，使用 IF 函数来判断获取
指定业务员的数据区域：

IF($M$2="业务员1",$B$3:$J$6,IF($M$2="业务员2",$B$7:$J$11,$B$12: 

$J$14))

这个 IF 嵌套的结果是指定业务员的数据区域，可以将其定义名称为“查找区域”，
这样公式就变得很简单了：

=IFERROR(

        VLOOKUP($M$4,查找区域,MATCH(L7,$B$2:$J$2,0)+IF($M$3= 

" 国内 ",0,4),0),

        " 没有数据 ")

图 1-19　复杂查找公式

定义名称的步骤如下。
步骤1在“公式”选项卡中单击“定义名称”命令按钮，如图 1-20 所示。打开“新建

名称”对话框，如图 1-21 所示。

	 			 																				

              图 1-20　“定义名称”命令按钮                  图 1-21　“新建名称”对话框

也可以单击“名称管理器”命令按钮，打开“名称管理器”对话框，如图 1-22 所示，
再单击对话框中的“新建”按钮，也会打开“新建名称”对话框。

图 1-22　“名称管理器”对话框

如果要定义很多名称，或者名称的引用公式很长，建议使用“名称管理器”对
话框来定义名称，因为定义好名称后，可以快速检查名称的引用位置是否正确。
步骤2在“名称”输入框中输入名称“查找区域”，在“引用位置”输入框中输入下面
的引用公式，如图 1-23 所示。

=IF($M$2="业务员1",$B$3:$J$6,IF($M$2="业务员2",$B$7:$J$11,$B$12: 

$J$14))

图 1-23　定义名称“查找区域”

步骤3单击“确定”按钮，就定义了名称“查找区域”，在“名称管理器”对话框中
就可以看到这个名称了，如图 1-24 所示。

图 1-24　定义的名称“查找区域”

1�3�2 名称应用案例：使用名称制作动态图表

在制作动态图表时，大多数情况下是设计辅助区域绘制图表，但在此情况下需
要定义动态名称，并利用定义的名称绘制图表。下面通过一个最简单的例子，说明
定义名称并绘制动态图表的基本方法和应用技能。

案例 1-6

图 1-25 是项目汇总表，A列项目会增加或减少。如何绘制一个可以随着项目增
加或减少而自动调整的动态图表？

图 1-25　示例数据，项目汇总表

定义“项目”和“数据”两个名称，引用公式分别如下，如图 1-26 所示。
名称“项目”：
=OFFSET($A$2,,,COUNTA($A$2:$A$100),1)

名称“数据”：
=OFFSET($B$2,,,COUNTA($A$2:$A$100),1)



01

02

03

04

05

06

07

08

10

09

第
1
章 

创
创
创
创创

创
创
创
创

创
创
创
创
创
创
创
创

17

图 1-26　定义的名称“项目”和“数据”

有了这两个名称，就可以使用名称绘制图表，绘制的图表如图 1-27 所示。

图 1-27　使用动态名称绘制的图表

这样，当项目增减时，图表就自动调整最新状态，如图 1-28 所示。

图 1-28　图表自动调整

1�4 使用数组公式解决复杂计算问题

在某些情况下，也会遇到对数组进行计算的场合，此时就需要设计数组公式来
解决一些复杂的计算问题。
数组公式，就是在公式中对数组进行计算，得到一个或多个结果的公式。在

Excel 中，数组公式是非常强大的，可以解决一些非常复杂的计算问题，以及做一些
复杂的数据处理和分析。

1�4�1 数组基本知识

所谓数组，就是把数据组合起来。在 Excel 工作表中，数组有一维数
组和二维数组两种。不论是一维数组还是二维数组，数组中的各个数据（又
称元素）需要用逗号或分号隔开。
数组中的各个元素，可以是同类数据，也可以是不同类型的数据。对于数字，

直接写上即可，但对于文本和日期，则需要用双引号括起来。
常见的数组有一维水平数组、一维垂直数组和二维数组。
一维水平数组的各个元素用逗号（,）隔开，相当于工作表的某行连续单元格数

据。例如，如果将数组 {1,2,3,4,5} 输入工作表，就是如图 1-29 所示的情形，从而数
组 {1,2,3,4,5} 就相当于单元格区域 B2:F2。

图 1-29　一维水平数组与工作表单元格区域的对应关系

一维垂直数组的各个元素用分号（;）隔开，相当于工作表的某列连续单元格数
据。例如，如果将数组 {1;2;3;4;5} 输入工作表，就是如图 1-30 所示的情形，从而数
组 {1;2;3;4;5} 就相当于单元格区域 B2:B6。

图 1-30　一维垂直数组与工作表单元格区域的对应关系

二维数组就是工作表中一个连续的矩形单元格区域。例如，如果将数组 {1,2,3,4,
5;6,7,8,9,10;11,12,13,14,15} 输入工作表，就是如图 1-31 所示的情形，从而数组 {1,2,
3,4,5;6,7,8,9,10;11,12,13,14,15} 就相当于单元格区域 B2:F4。

图 1-31　二维数组与工作表单元格区域的对应关系

Excel 中，数组公式是非常强大的，可以解决一些非常复杂的计算问题，以及做一些
复杂的数据处理和分析。

1�4�1 数组基本知识

所谓数组，就是把数据组合起来。在 Excel 工作表中，数组有一维数
组和二维数组两种。不论是一维数组还是二维数组，数组中的各个数据（又
称元素）需要用逗号或分号隔开。
数组中的各个元素，可以是同类数据，也可以是不同类型的数据。对于数字，

直接写上即可，但对于文本和日期，则需要用双引号括起来。
常见的数组有一维水平数组、一维垂直数组和二维数组。
一维水平数组的各个元素用逗号（,）隔开，相当于工作表的某行连续单元格数

据。例如，如果将数组 {1,2,3,4,5} 输入工作表，就是如图 1-29 所示的情形，从而数
组 {1,2,3,4,5} 就相当于单元格区域 B2:F2。

图 1-29　一维水平数组与工作表单元格区域的对应关系

一维垂直数组的各个元素用分号（;）隔开，相当于工作表的某列连续单元格数
据。例如，如果将数组 {1;2;3;4;5} 输入工作表，就是如图 1-30 所示的情形，从而数
组 {1;2;3;4;5} 就相当于单元格区域 B2:B6。

图 1-30　一维垂直数组与工作表单元格区域的对应关系

二维数组就是工作表中一个连续的矩形单元格区域。例如，如果将数组 {1,2,3,4,
5;6,7,8,9,10;11,12,13,14,15} 输入工作表，就是如图 1-31 所示的情形，从而数组 {1,2,
3,4,5;6,7,8,9,10;11,12,13,14,15} 就相当于单元格区域 B2:F4。

图 1-31　二维数组与工作表单元格区域的对应关系

1�4�2 数组公式基本知识

所谓数组公式，就是对数组进行计算的公式。数组公式有以下特征：
 ● 单击数组公式所在的任一单元格，就可以在公式编辑栏中看到公式

前后出现的大括号“{ }”，如果在公式编辑栏中单击，大括号就会消

失。这个大括号是自动显示出来的，表示这个是数组公式，千万不要添加这个

大括号。

 ● 如果是在几个单元格中输入数组公式，那么每个单元格中的公式是完全相同的。

 ● 必须同时按组合键 Ctrl+Shift+Enter 才能得到数组公式（此时可以在编辑栏中

看到公式的前后有一对大括号），否则，如果只按 Enter 键，那样得到的是普通 
公式。

 ● 公式中必定有单元格区域的引用，或者必定有数组常量。

 ● 不能单独对数组公式所涉及的单元格区域中的某一个单元格进行编辑、删除或

移动等操作。

 ● 不能在合并单元格中输入数组公式。

在工作表中输入数组公式必须遵循一定的方法和步骤。尽管数组公式返回值可
以是多个或一个，但输入数组公式的基本方法是一样的，最后都必须按组合键 Ctrl+	
Shift+Enter，唯一的区别是选取单元格的不同：

 ● 返回多个结果的数组公式要选取连续的单元格区域。

 ● 返回一个结果的数组公式仅选取一个单元格。

数组公式的输入方法和基本步骤如下。
步骤1选定某个单元格或单元格区域。
如果数组公式返回一个结果，单击需要输入数组公式的某个单元格。
如果数组公式将返回多个结果，则要选定需要输入数组公式的单元格区域。

步骤2输入数组公式。
步骤3按组合键 Ctrl+Shift+Enter，Excel 自动在公式的两边显示大括号 {	}。
当需要对数组公式进行编辑时，可以按照下列的方法来操作：
如果是返回一个结果的数组公式，那么就仅选取该单元格，编辑公式，然后按

组合键 Ctrl+Shift+Enter。
如果是返回多个结果的数组公式，那么就要选择该数组公式中的全部单元格，

编辑公式，然后按组合键 Ctrl+Shift+Enter。

1�4�3 数组公式应用案例：计算前N大数据之和

了解数组及数组公式的基本知识后，下面介绍一个数组公式的具体应用案例。

案例 1-7

图 1-32 中的示例，要求计算前 5大客户销售额合计数。这里，前 5大客户销售
额合计公式如下：

{=SUM(LARGE(B2:B18,{1,2,3,4,5}))}

在这个公式中，使用了数组 {1,2,3,4,5} 来构建一个连续的序号，以便使用
LARGE 函数分别取出前 5大数据。因此，LARGE(B2:B18,{1,2,3,4,5})	的结果就是前 5
大数据构成的数组 {21029,11055,9949,8241,5945}，然后再使用 SUM函数将这 5个
数字相加，就是前 5大客户销售额合计数了。

图 1-32　数组公式经典应用

思考一下，如果要计算前N大客户销售额合计数，这里N是一个变量，如何设
计公式呢？
如图 1-33 所示，假设单元格 F2 指定 N的数值，那么可以在单元格 F3 中输入如

下数组公式：
=SUM(LARGE(B2:B18,ROW(INDIRECT("1:"&F2))))

下面是这个公式的计算逻辑。
 ● "1:"&F2 是构建一个 1 到几的字符串。例如，单元格 F2 指定数字 5，那么

"1:"&F2 就是字符串 "1:5"。
 ● 使 用 INDIRECT 函 数 将 这 个 字 符 串 "1:5" 转 换 为 对 第 1~5 行 的 引 用：

INDIRECT("1:"&F2)。
 ● 使用 ROW 函数提取出第 1~5 行的行号：ROW(INDIRECT("1:"&F2))，这个表

达式就生成了一个 1~5 的连续序号：{1;2;3;4;5}。
 ● 使用 LARGE 函数提取出前 5 大数据，LARGE(B2:B18,ROW(INDIRECT("1:"&
F2)))，这个表达式的结果就构成一个数组：{21029;11055;9949;8241;5945}。



01

02

03

04

05

06

07

08

10

09

第
1
章 

创
创
创
创创

创
创
创
创

创
创
创
创
创
创
创
创

21

 ● 最后使用 SUM 函数将前 5 大数据的数组进行求和。

图 1-33　计算前 N 大数据之和

指定单元格 F2 中的任意数字，就可以立即算出前N大客户销售额合计数。如图
1-34 所示，就是计算前 10 大客户销售额的合计数。

图 1-34　前 10 大客户销售额

此时，ROW函数的结果是数组 {1;2;3;4;5;6;7;8;9;10}，因此，LARGE 函数的结果
是数组 {21029;11055;9949;8241;5945;4035;1498;1359;1350;729}。

�ᐺ本节知识回顾与测验

1. 定义名称要注意哪些问题？

2. 如何对任意单元格、单元格区域、常量等定义名称？

3. 定义的名称字符之间为什么不能有空格？

4. 定义名称、查看编辑引用公式，最方便的方法是什么？

5. 如果定义了一个名称“税率”，其引用公式为“=0.25”，那么在公式中，如果

输入“税率”，如公式“=C5* 税率”，这个“税率”是什么？

6. 什么是数组？一维数组、二维数组与工作表单元格区域有什么对应关系？

7. 如何在一个单元格中输入数组公式？如何在一个单元格区域输入数组公式？

8. 要完成数组公式的输入，必须按什么键？

9. 如何查看编辑数组公式？

10. 如何计算一个单元格区域的 10 个最大数字之和？如何计算一个单元格区域

的最小 10 个数字之和？


